CONVERSION OF UTM COORDINATES TO GEOGRAPHIC COORDINATES FOR SAS/GRAPH® SOFTWARE PROC GMAP DISPLAY

David C. Chojnacky, Intermountain Research Station
Ronald P. Tymcio, Intermountain Research Station

ABSTRACT: The Universal Transverse Mercator (UTM) grid is used to reference all forest data collected in the Rocky Mountain States by the USDA Forest Service, Forest Survey Project. Display of these data with SAS/GRAPH® mapping procedures requires a conversion of UTM coordinates to geographic coordinates (latitude and longitude). In this paper a SAS® DATA step is given for conversion of UTM to geographic coordinates. Also given are examples of Forest Survey maps using the SAS/GRAPH with converted UTM data.

INTRODUCTION

The Universal Transverse Mercator (UTM) grid is a map reference system for almost the entire world (USDD 1973). Only the North and South Poles are excluded. The grid consists of geographical strips or zones 6 degrees in width (fig. 1).



Figure 1--UTM grid zones for the United States.

Map points are identified within each zone by "northing" and "easting" UTM coordinates on a metric scale (fig. 2).



Figure 2--The UTM "northing" coordinate identifies vertical direction and the "easting" coordinate identifies horizontal direction.

UTM coordinates are planimetric and are continuous except for small breaks in the easting scale between zones. UTM coordinates are the fundamental map reference system used by the USDA Forest Service, Forest Survey Unit in the Rocky Mountain States (fig. 3).

Figure 3--The Rocky Mountain States.

All forest land sampling done by Forest Survey is referenced to the UTM system. An initial 1 000-meter UTM grid is marked on U.S. Geologic Survey maps to identify forested land and ownership categories. Then field locations are subsampled from the 1 000-meter grid sample for ground observation and measurement. By sampling from a UTM grid, Forest Survey samples are equal map distance apart (within grid zones), field sample locations are easily identified on U.S. Geologic Survey topographic maps, and resulting survey data are labeled in metric units.

Although UTM coordinates have many advantages for mapping and inventory of forest and range lands in the Rocky Mountain States, they are not compatible with SAS mapping procedures (SAS 1985). The GMAP procedure in SAS/GRAPH is based on geographic coordinates (latitude and longitude expressed in radians), not UTM coordinates. Fortunately, a conversion between geographic and UTM coordinates exists, but it is somewhat complicated.

The purpose of this paper is to document the conversion of UTM coordinates to geographic coordinates in a SAS DATA step and display resulting data with GMAP procedure using the ANNOTATE= Data Sets feature.

CONVERSION OF UTM COORDINATES

The algorithm for conversion of UTM coordinates to geographic coordinates was taken from documentation supplied by Lloyd (1985), which was based on Claire's (1968) work. For a given map point, the algorithm requires UTM grid zone and northing and easting coordinates as inputs in order to give latitude and longitude as outputs. The Clark 1866 spheroid is used by the algorithm for the assumed size and shape of the earth. The algorithm, as given in this paper, is applicable to the northern hemisphere only, but easily can be modified for the southern hemisphere by adjusting the northing:

$$Y' = 10{,}000{,}000 - Y$$

where:

Y = northing coordinate in southern hemisphere

Y' = adjusted northing coordinate.

The mathematics of the conversion only involve algebra, but the conversion itself is based on complex geometric relationships. Further explanation is not given here but can be found in USDD (1973). Given below is a SAS DATA step including brief comments for the conversion of UTM coordinates in the northern hemisphere. Inputs for the DATA step are UTM grid zone (2-digit integer), easting (6-digit integer), and northing (7-digit integer). Also state and county (in FIPS code) are included for later merging with SAS/GRAPH map data sets.

```
*****
*SAS PROGRAM FOR CONVERSION OF UTM EASTING AND NORTHING
*TO LATITUDE (LATRAD) AND LONGITUDE (LONGRAD) EXPRESSED
*IN RADIANS -- USING VERSION 5.03 UNDER AOS/VS
*****;

LIBNAME DD2 ':D3:SASLIB:GMAP';

DATA RADIANS (KEEP=STATE COUNTY EASTING NORTHING
                   LATRAD LONGRAD);

INFILE UTMS MISSOVER RECFM=D;
  INPUT STATE COUNTY GRIDZONE EASTING NORTHING

***DEFINE VARIABLES***;
  X=EASTING;
  Y=NORTHING;
  ZONE=GRIDZONE;

***DEFINE PI***;
  PI=3.141592653589793;

***RADIUS OF THE EQUATOR IN METERS***;
  A=6378206.4;

***ECCENTRICITY SQUARED***;
  ESQ=.006768657997;

***MINOR ECCENTRICITY SQUARED***;
  EM2=ESQ/(1-ESQ);

***SCALE FACTOR OF THE CENTRAL MERIDIAN***;
  K0=.9996;

***FALSE EASTING***;
  C=500000.0;

***RECTIFYING LATITUDE***;
  V=.0000001571128261;
  RL=V*Y;
  CRL=COS(RL);

***CALCULATING LATITUDE OF FOOT OF PERPENDICULAR
   FROM POINT TO CENTRAL MERIDIAN (IN RADIANS)***;
  G=.0000002457956882;
  D=.005078649674;
  F=.0000300245646;
  FP=RL + ((D + (F*CRL**2)
     + (G*CRL**4))*(SQRT(1.0-CRL**2))*CRL);

***COSINE OF LATITUDE OF FOOT OF PERPENDICULAR***;
  COSFP=COS(FP);

***COSINE OF LATITUDE OF FOOT OF PERPENDICULAR
   SQUARED***;
  CFP2=COSFP**2;
  ETA=1 + (EM2*CFP2);

***DIFFERENCE BETWEEN EASTING INPUTED AND
   FALSE EASTING***;
  XMC=X-C;
  XAK=((1-ESQ)/(K0*K0))*(XMC/A)**2;

***DEGREES IN ONE RADIAN FOR CALCULATING
   CENTRAL MERIDIAN***;
  RADDEG=180.0/PI;

***CALCULATIONS FOR LATITUDE***;
  *FIRST CALCULATION*;
  LAT1=ETA*ETA*TAN(FP)/2.0;

  *SECOND CALCULATION*;
  LAT2=(2.0 - 6.0*EM2 + 3.0/CFP2
       + (12.0*EM2 - 9.0*EM2**2)*CFP2
       + 6.0*EM2**2*CFP2)/12.0;

  *THIRD CALCULATION*;
  LAT3= ETA*(16.0 - 72.0*EM2
       + 45.0/CFP2**2 - 45.0*EM2/CFP2
       + 244.0*EM2*CFP2)/360.0;

***CALCULATING LATITUDE IN RADIANS***;
  LATRAD=FP - LAT1*XAK*(1.0 - LAT2*XAK + LAT3*XAK**2);

***CALCULATING LONGITUDE***;
  *FIRST CALCULATION*;
   LONG1=SQRT(ETA)/COSFP;

  *SECOND CALCULATION*;
   LONG2=ETA*(-1.0 + 2.0/CFP2 + EM2*CFP2)/6.0;

  *THIRD CALCULATION*;
   LONG3=ETA*(1.0 + 8.0*EM2 + 24.0/CFP2**2 - 20.0/CFP2
        - 2.0*EM2*CFP2)/120.0;

***CALCULATING CENTRAL MERIDIAN IN RADIANS FOR
   GRID ZONE ENTERED***;
  CM=(183.0 - 6.0*ZONE)/RADDEG;

***CALCULATING LONGITUDE IN RADIANS***;
 *NOTE: ADD DIFFERENCE IF EASTING LESS THAN FALSE EASTING
   SUBTRACT DIFFERENCE IF EASTING GREATER THAN FALSE
   EASTING OF THE GRID ZONE;
   IF XMC LT 1 THEN
      LONGRAD=CM + SQRT(XAK)*LONG1*(1.0 - XAK*
                          LONG2 + XAK**2 * LONG3);
   IF XMC GT 1 THEN
      LONGRAD=CM - SQRT(XAK)*LONG1*(1.0 - XAK*
                          LONG2 + XAK**2 * LONG3);
RUN;
```

SAS UTM MAPPING APPLICATIONS

Several maps were made to illustrate the usefulness of the UTM conversion algorithm when combined with the mapping features of SAS/GRAPH. The UTM data (converted to latitude and longitude) were combined and projected with the SAS map data set to assure uniform scaling. The projected UTM data were then separated from the SAS map data set. The projected UTM data were included in the GMAP procedure for final mapping using ANNOTATE= Data Sets. A similar example of this logic is given in the SAS/GRAPH User's Guide (SAS 1985, p. 316-317).

Figure 4 illustrates 5 000-meter UTM grid points corresponding to several ownerships overlaid onto a Wyoming counties map. Five different public

Figure 4--Five Wyoming public ownerships sampled on 5 000-meter UTM grid.

ownerships are illustrated for Wyoming. The map shows a slight problem in using special symbols with FUNCTION='LABEL' to annotate the ownerships on the map. Notice in figure 4 ownership classes below the southern Wyoming border and a small gap of no ownerships below the northern Wyoming border. There should be no ownership symbols outside the Wyoming border and National Park and National Forest ownerships should be adjacent to the northern border. These problems apparently occurred because ANNOTATE's label function centers within a graphics cell, instead of centering exactly on the X an Y coordinates (SAS 1985, p. 124). Had we used ANNOTATE's symbol function (SAS 1985, p. 130) instead of the label function this problem might not have occurred. However this speculation was not tested because of a "bug" in our AOS/VS version of SAS/GRAPH (beta test version 5.03).

Also illustrated in figure 4 is the effect of a UTM zone break between zones 12 and 13 (see fig. 1) for the BLM ownership. Zone breaks are a known nuisance when selecting survey samples on a fixed UTM coordinate grid, but maps such as figure 4 will help us make a sampling correction to deal with this problem in the future.

Because the SAS code for generating figure 4 is representative of our concept for mapping UTM data, it is given below:

```
*****
*SAS PROGRAM FOR SUGI12 PAPER--OWNERSHIP MAP
*****;

GOPTIONS DEVICE=HP7475A;

LIBNAME DD1 ':D3:SASLIB:GMAP:SUGI12';

DATA PI; SET DD1.WY_PI;
  O=OWNER;
  IF O EQ 9 OR O EQ 11 OR O EQ 12 OR O EQ 13 OR
     O EQ 15;

  LENGTH FUNCTION $ 8 STYLE $ 8 TEXT $ 25 COLOR $ 8
         POSITION $ 1;
  FUNCTION='LABEL';
    XSYS='2';
    YSYS='2';
    WHEN='A';
    X=LONGRAD;
    Y=LATRAD;
    POSITION='5';
    STYLE='SPECIAL';
    COLOR='BLACK3';

    IF O EQ 11 THEN TEXT='K';     *NATIONAL FOREST;
    IF O EQ  9 THEN TEXT='H';     *NATIONAL PARK;
    IF O EQ 12 THEN TEXT='L';     *BLM;
    IF O EQ 13 THEN TEXT='$';     *INDIAN;
    IF O EQ 15 THEN TEXT='M';     *STATE;

    IF O EQ 9 OR O EQ 13 OR O EQ 15 THEN SIZE=.8;
    IF O EQ 11                      THEN SIZE=1.1;

  COUNTY = -1;  *** DUMMY COUNTY FOR ANNOTATE TEXT;
RUN;
```

```
DATA LEGEND;
  LENGTH TEXT $ 25;

  FUNCTION='LABEL';
    XSYS='3';
    YSYS='3';
    COLOR='BLACK3';
    POSITION='0';
    Y=01; X=14;
      STYLE='SPECIAL'; TEXT='K';                    OUTPUT;
    Y=.;  X=.;
      STYLE='DUPLEX';  TEXT=' Nat''l Forest';       OUTPUT;
      STYLE='SPECIAL'; TEXT='   H';                 OUTPUT;
      STYLE='DUPLEX';  TEXT=' Nat''l Park';         OUTPUT;
      STYLE='SPECIAL'; TEXT='   L';                 OUTPUT;
      STYLE='DUPLEX';  TEXT=' Bureau of Land Mgmt';OUTPUT;
      STYLE='SPECIAL'; TEXT='   $';                 OUTPUT;
      STYLE='DUPLEX';  TEXT=' Indian Reservation'; OUTPUT;
      STYLE='SPECIAL'; TEXT='   M';                 OUTPUT;
      STYLE='DUPLEX';  TEXT=' State';               OUTPUT;
RUN;

DATA SASMAP; SET DD1.WY_CNTY PI; RUN;

PROC GPROJECT DATA=SASMAP OUT=COMBINED;
  ID COUNTY;
RUN;

DATA MAP_PTS PI_PTS; SET COMBINED;
  IF COUNTY EQ -1 THEN OUTPUT PI_PTS;
                  ELSE OUTPUT MAP_PTS;
RUN;

DATA PI_PTS; SET PI_PTS LEGEND;
RUN;

DATA FAKE (KEEP=COUNTY NUMBER); SET MAP_PTS;
  BY COUNTY;
  IF FIRST.COUNTY;
  NUMBER=1;
RUN;

PROC GMAP MAP=MAP_PTS DATA=FAKE;
  ID COUNTY;
  CHORO NUMBER / COUTLINE=BLACK7 DISCRETE
                 NOLEGEND ANNOTATE=PI_PTS;
  PATTERN1 V=E;
  TITLE;
  FOOTNOTE1;
  FOOTNOTE3;
RUN;
```

Additional resource maps for sections of Wyoming are shown in figures 5 and 6.



Figure 5--Yellowstone National Park's forest types and lakes sampled on a 5 000-meter UTM grid.

In both cases UTM data sampled on 5 000-meter grid is overlaid onto a portion of the Wyoming counties map available in SAS.



Figure 6--Wood volume for all ownerships except National Forest in the Black Hills region of northeast Wyoming sampled on a 5 000-meter UTM grid.

These maps illustrate the usefulness of a spatial display of forest survey data. In the past, Forest Survey data presentation was limited to statistical tables or charts.

Our final mapping application of UTM data involved an attempt to draw boundaries around a block of UTM data points. We selected the Wind River Indian Reservation as a test case. Identification of the boundary points (fig. 7) was straightforward. However, we had difficulty joining the points with a solid line.



Figure 7--Boundary perimeter points for Wyoming's Wind River Indian Reservation sampled on a 5 000-meter UTM grid.

In figure 8 the northeast and northwest corner boundaries double back.



Figure 8--Boundary drawn for Wyoming's Wind River Indian Reservation.

This was due to our inability to easily sort the boundary points in the correct order for certain types of concave corners. However, our SAS boundary program, given below, is a reasonable approximation for crude mapping purposes:

```
*****
*SAS PROGRAM FOR IDENTIFICATION OF WIND RIVER INDIAN
*    RESERVATION BOUNDARY POINTS FOR 5 000-METER UTM GRID
*****;

GOPTIONS DEVICE=HP7475A;

LIBNAME DD1 ':D3:SASLIB:GMAP:SUGI12';

DATA WR_5000M; SET DD1.WR_5000M;    ***5 000-METER DATA FILE;
RUN;

PROC SORT DATA=WR_5000M; BY DESCENDING NORTHING EASTING;
RUN;

DATA EAST; SET WR_5000M; BY DESCENDING NORTHING;
  RETAIN KNT 0;
  IF LAST.NORTHING THEN DO;
     KNT=KNT+1;
     SIDE=1;
     OUTPUT EAST;
  END;
RUN;

PROC SORT DATA=WR_5000M; BY DESCENDING EASTING NORTHING;
RUN;

DATA SOUTH; SET WR_5000M; BY DESCENDING EASTING;
  RETAIN KNT 0;
  IF FIRST.EASTING THEN DO;
     KNT=KNT+1;
     SIDE=2;
     OUTPUT SOUTH;
  END;
RUN;

PROC SORT DATA=WR_5000M; BY NORTHING EASTING;
RUN;

DATA WEST; SET WR_5000M; BY NORTHING;
  RETAIN KNT 0;
  IF FIRST.NORTHING THEN DO;
     KNT=KNT+1;
     SIDE=3;
     OUTPUT WEST;
  END;
RUN;

PROC SORT DATA=WR_5000M; BY EASTING DESCENDING NORTHING;
RUN;

DATA NORTH; SET WR_5000M; BY EASTING;
  RETAIN KNT 0;
  IF FIRST.EASTING THEN DO;
     KNT=KNT+1;
     SIDE=4;
     OUTPUT NORTH;
  END;
RUN;

PROC SORT DATA=SOUTH; BY DESCENDING NORTHING EASTING;
RUN;

PROC SORT DATA=WEST; BY DESCENDING NORTHING EASTING;
RUN;

PROC SORT DATA=NORTH; BY DESCENDING NORTHING EASTING;
RUN;

DATA WR_BNDRY; MERGE EAST SOUTH WEST NORTH;
               BY DESCENDING NORTHING EASTING;
RUN;

PROC SORT; BY SIDE KNT;
RUN;

DATA WR_BNDRY; SET WR_BNDRY;
  LENGTH FUNCTION $ 8;

  RETAIN I 0;
  I=I+1;
  IF I EQ 1 THEN FUNCTION='MOVE';
            ELSE FUNCTION='DRAW';
   XSYS='2';
   YSYS='2';
   WHEN='A';
   LINE=1;
   SIZE=40;
   COLOR='BLACK7';
   X=LONGRAD;
   Y=LATRAD;
   COUNTY=-1;   *** DUMMY COUNTY FOR ANNOTATE TEXT;
RUN;

DATA WR_MAP; SET DD1.WR_CNTY WR_BNDRY; RUN;

PROC GPROJECT DATA=WR_MAP OUT=WR_MAP;
  ID COUNTY;
RUN;

DATA MAP_PTS WR_BNDRY; SET WR_MAP;
  IF COUNTY EQ -1 THEN OUTPUT WR_BNDRY;
  IF COUNTY GT  0 THEN OUTPUT MAP_PTS;
RUN;

PROC GMAP MAP=MAP_PTS DATA=MAP_PTS;
  ID COUNTY;
  CHORO COUNTY / COUTLINE=BLACK7 DISCRETE NOLEGEND
                 ANNOTATE=WR_BNDRY;
  PATTERN1 V=E;
  PATTERN2 V=E;
  TITLE1 H=3;
  TITLE2 H=3;
RUN;
```

CONCLUSION

We feel this paper has illustrated the usefulness of combining UTM referenced data with the powerful features of SAS/GRAPH to improve Forest Survey data reporting. The previous examples represent just the beginning of a new era of Forest Survey data display.

REFERENCES

U.S. Department of Defense, Army. Universal transverse mercator grid. Technical Manual TM 5-241-8. Washington, DC: U.S. Department of Defense, Army, Headquarters Department of the Army; 1973. 64 p.

Lloyd, W.R. [Documentation of UTM easting, northing conversion to latitude, longitude] 1985 December 12. Located at: U.S. Department of Interior, Geological Survey, National Mapping Division, Menlo Park, CA.

Claire, C.N. State plane coordinates by automatic data processing. Publication 62-4. Washington, DC: U.S. Department of Commerce, Coast and Geodetic Survey, 1968. 68 p.

SAS Institute, Inc. SAS/GRAPH user's guide, version 5 edition. Cary, NC: SAS Institute, Inc., 1985. 596 pp.

Paper presented at SUGI 12, Dallas, TX, February 8-11, 1987. David C. Chojnacky is a Research Forester and Ronald P. Tymcio is a Computer Programmer; both are with Intermountain Research Station, Forest Service, U.S. Department of Agriculture, Ogden, UT 84401. Telephone: (801) 625-5388